
Po tych zajęciach

• odróżniasz kontekst filtra od kontekstu wiersza i wiesz, który liczy twoją miarę

• piszesz miary z VAR, DIVIDE i CALCULATE, które liczą dobrze także w wierszu sumy

• używasz iteratorów i pięciu funkcji, które zamieniają liczbę w tekst, kolor i ranking

• sprawdzasz miarę, zanim uwierzysz wynikowi — także miarę napisaną przez AI

Gdzie jesteśmy w projekcie

Zajęcia 3 · model

Gwiazda: fakty w środku, 
wymiary wokół, relacje jeden 
do wielu. Model mówi, którędy 
płyną filtry.

Dziś · miary

Miara liczy wynik w każdej 
komórce raportu. Dziś napiszesz 
ich 25 — od SUM do WINDOW.

Zajęcia 5 · CP2

Oddajesz model i miary 
swojego projektu jako PBIP. Dziś 
zbierasz wzorce, które tam 
przeniesiesz.

Miary z dzisiejszych ćwiczeń to też wzorce na obronę: karty A1–A4 sprawdzają dokładnie te umiejętności.

01
Miara liczy się tam, gdzie ją postawisz



Kolumna obliczeniowa czy miara?

Kolumna obliczeniowa Miara

Kiedy liczy raz, przy odświeżeniu danych przy każdym filtrze i kliknięciu

Gdzie jest wynik w tabeli — zajmuje pamięć nigdzie — liczy się na żądanie

Co widzi jeden wiersz: kontekst wiersza filtry komórki: kontekst filtra

Przykład Marża wiersza = [Wartość sprzedaży] - [Koszt 
całkowity]

Marża % = DIVIDE([Marża PLN], [Sprzedaż PLN])

Użyj, gdy wynik ma trafić na oś, do fragmentatora albo relacji wynik ma się sumować, dzielić i reagować na filtry

W projekcie domyślnie piszesz miary. Kolumnę obliczeniową dodajesz, gdy potrzebujesz nowej kategorii, np. przedziału cen.

Kontekst filtra: co widzi jedna komórka
Każda komórka tabeli, słupek i karta to osobne obliczenie miary. Filtry komórki decydują, które wiersze faktów policzy.

Sprzedaż PLN, mln zł 2023 2024 2025

Komputery 90,8 65,2 92,2

Telewizory i wideo 58,7 50,5 60,6

Telefony 46,6 49,1 51,3

AGD 22,7 24,4 25,4

… 4 kolejne kategorie

Suma 284,8 262,0 303,7

KOMÓRKA KOMPUTERY × 2024

• filtr z wiersza: 'Produkt'[Kategoria] = "Komputery"

• filtr z kolumny: 'Kalendarz'[Rok] = 2024

• do tego fragmentatory oraz filtry strony i raportu

Relacje przenoszą oba filtry do tabeli Sprzedaż. Z 57 222 
wierszy zostaje 3 193 — i tylko je sumuje SUM: 
65 185 656 zł.

A WIERSZ SUMY?

Ma tylko filtr roku, więc pokazuje całą sieć: 262,0 mln 
zł w 2024 r.

Kontekst wiersza: liczysz jeden wiersz naraz
Kolumna obliczeniowa i iterator widzą bieżący wiersz tabeli. Nie widzą filtrów raportu.

KOLUMNA OBLICZENIOWA W TABELI SPRZEDAŻ

Marża wiersza =

'Sprzedaż'[Wartość sprzedaży]

- 'Sprzedaż'[Koszt całkowity]

Power BI liczy wyrażenie dla każdego z 57 222 wierszy przy 
odświeżeniu i zapisuje wynik w tabeli.

MIARA Z ITERATOREM SUMX

Sprzedaż brutto PLN =

SUMX(

'Sprzedaż',

'Sprzedaż'[Ilość] * 'Sprzedaż'[Cena jednostkowa]

)

SUMX tworzy kontekst wiersza tylko na czas liczenia: mnoży 
w każdym wierszu, potem sumuje. Nic nie zapisuje.

PUŁAPKA

Kontekst wiersza nie filtruje. SUM('Sprzedaż'[Ilość]) w kolumnie obliczeniowej zwróci w każdym wierszu tę samą sumę całej tabeli: 
619 376.

Trzy nawyki dobrej miary

Nazwa mówi, co liczy

Sprzedaż PLN, nie Miara 1. 
Jednostka w nazwie, miara 
w tabeli _Miary, w folderze 
tematu.

Format przy mierze

Ustaw format raz: Narzędzia 
miar → Format. #,0 dla złotych, 
0,0% dla udziałów. Każdy 
wykres go przejmie.

Dobrze także w sumie

Wiersz sumy to też kontekst 
filtra. Sprawdzaj go zawsze —
tam wychodzą błędy udziałów, 
rankingów i średnich.

Opis miary (pole Opis we właściwościach) widać w dymku w panelu Dane. Wypełnij go — kryteria oceny miar go wymagają.

02
Miara z miar: VAR i DIVIDE



Ćw. 0 · Otwórz plik z tabelą miar 5 min

1 Pobierz z Moodle INWD_dax_cwiczenia.zip i rozpakuj do C:\ — pliki trafią 
do C:\INWD.

2 Otwórz C:\INWD\INWD_dax_cwiczenia.pbip i kliknij „Odśwież teraz”.

3 W panelu Dane znajdź tabelę _Miary: folder Podstawowe, a w nim 
Sprzedaż PLN i Sprzedaż szt.

4 Otwórz stronę „Stanowisko testowe”. Tu sprawdzisz każdą nową miarę.

Wskazówka: Tabela miar to pusta tabela z jedną ukrytą kolumną. Power BI pokazuje ją na 
górze panelu Dane.

Stanowisko testowe: trzy tabele w trzech ziarnach i dwa fragmentatory.

Tabela miar we własnym projekcie
W projekcie na zaliczenie zrób ją od razu. Miary nie zgubią się wśród kolumn.

• Narzędzia główne → Wprowadź dane → nazwa tabeli _Miary → Załaduj

• Istniejącą miarę przeniesiesz tak: zaznacz ją → Narzędzia miar → Tabela główna → _Miary

• Ukryj kolumnę Kolumna1 — tabela zostanie tylko z miarami i trafi na górę panelu Dane

• Foldery: widok modelu → zaznacz miary → Właściwości → Folder wyświetlania

VAR i DIVIDE: miara, którą da się przeczytać

Z VAR

Marża % =

VAR _sprzedaz = [Sprzedaż PLN]

VAR _koszt = [Koszt PLN]

RETURN

DIVIDE(_sprzedaz - _koszt, _sprzedaz)

BEZ VAR — DZIAŁA, ALE CZYTA SIĘ GORZEJ

DIVIDE([Sprzedaż PLN] - [Koszt PLN],

[Sprzedaż PLN])

• VAR nadaje nazwę wynikowi pośredniemu. Liczy go raz, 
w kontekście miejsca, w którym stoi.

• RETURN mówi, co miara zwraca. Chcesz sprawdzić krok 
pośredni? Zwróć na chwilę _koszt.

• DIVIDE(a, b) zamiast a / b: gdy b to zero albo pusto, 
dostajesz pustą komórkę zamiast błędu.

• Nazwy zmiennych bez spacji i polskich znaków: 
_sprzedaz, _koszt.

Ćw. 1 · VAR, DIVIDE i format 10 min

1 W _Miary utwórz Koszt PLN = SUM('Sprzedaż'[Koszt całkowity]) i Marża PLN = [Sprzedaż PLN] - [Koszt PLN].

2 Utwórz Marża % z kodu poniżej. Format: Narzędzia miar → Procent, jedno miejsce po przecinku.

3 Dodaj Średnia cena PLN = DIVIDE([Sprzedaż PLN], [Sprzedaż szt.]).

4 Przeciągnij cztery miary do tabeli „Według roku” na stanowisku testowym.

Marża % =

VAR _sprzedaz = [Sprzedaż PLN]

VAR _koszt = [Koszt PLN]

RETURN

DIVIDE(_sprzedaz - _koszt, _sprzedaz)

PYTANIE

Marża % w wierszu sumy to 27,8%, a średnia z trzech lat — 27,7%. Skąd ta różnica?

Rezultat ćwiczenia 1

SPRAWDŹ U SIEBIE

• Marża %: 28,6% → 25,9% → 
28,7%

• Średnia cena: 1 481 → 1 250 
→ 1 396 zł

• suma: marża 27,8%, cena 
1 373 zł

• Marża PLN 2024: 67,9 mln zł

Plik INWD_dax_rozwiazanie.pbip, strona „Ćw. 1”. Tytuł strony to wniosek: marża i cena spadły w tym samym roku.

03
CALCULATE: zmieniasz kontekst filtra



CALCULATE: trzy ruchy na filtrach
CALCULATE(wyrażenie, filtr, …) liczy wyrażenie w zmienionym kontekście filtra.

Dodaj filtr

CALCULATE(

[Sprzedaż PLN],

'Kanał'[Typ kanału]

= "Online"

)

Filtr na kolumnie zastępuje filtr tej 
kolumny. Rok, kategoria i region 
zostają.

Usuń filtr

CALCULATE(

[Sprzedaż PLN],

REMOVEFILTERS('Produkt')

)

Zdejmuje wszystkie filtry z tabeli 
Produkt, także z fragmentatora. To 
mianownik udziału w całości.

Zostaw wybór użytkownika

CALCULATE(

[Sprzedaż PLN],

ALLSELECTED('Produkt')

)

Zdejmuje filtr wiersza tabeli, ale 
zostawia to, co wybrano we 
fragmentatorze.

REMOVEFILTERS i ALL działają w CALCULATE tak samo. REMOVEFILTERS mówi wprost, co robi — dlatego go używamy.

Dwa mianowniki: REMOVEFILTERS kontra ALLSELECTED
Fragmentator: Komputery, Telefony, Telewizory i wideo. Ta sama sprzedaż, dwa różne udziały.

Kategoria Sprzedaż PLN Udział w całości % Udział w wybranych %

Komputery 248 110 544 29,2% 43,9%

Telewizory i wideo 169 782 276 20,0% 30,1%

Telefony 146 954 285 17,3% 26,0%

Suma 564 847 105 66,4% 100,0%

REMOVEFILTERS('Produkt') mianownik 850 573 496: cała sieć

ALLSELECTED('Produkt') mianownik 564 847 105: trzy wybrane kategorie

KTÓRY WYBRAĆ?

• O mianowniku decyduje 
pytanie odbiorcy.

• „Jaką część całego 
biznesu dają 
komputery?” —
REMOVEFILTERS.

• „Jaką część tego, co 
oglądam?” —
ALLSELECTED.

KEEPFILTERS: dołóż filtr, zamiast go podmieniać

Komputery PLN = CALCULATE([Sprzedaż PLN], 'Produkt'[Kategoria] = "Komputery")

Komputery KEEPFILTERS PLN = CALCULATE([Sprzedaż PLN], KEEPFILTERS('Produkt'[Kategoria] = "Komputery"))

Kategoria Sprzedaż PLN Komputery PLN Komputery KEEPFILTERS PLN

Komputery 248 110 544 248 110 544 248 110 544

Telewizory i wideo 169 782 276 248 110 544

Telefony 146 954 285 248 110 544

• Zwykły filtr w CALCULATE 
zastępuje filtr kategorii 
z wiersza — w każdym wierszu 
liczysz komputery.

• KEEPFILTERS łączy go z filtrem 
wiersza. Telefony i komputery 
naraz to zbiór pusty.

Dla chętnych w ćw. 2: postaw obie miary obok siebie w tabeli „Według kategorii” na stanowisku testowym.

Ćw. 2 · CALCULATE: trzy wzorce 15 min

1 Utwórz Sprzedaż online PLN i Udział online % (kod 1). Sprawdź je w tabeli „Według roku”.

2 Utwórz Sprzedaż wszystkich produktów PLN i Udział w całości % (kod 2).

3 Utwórz Udział w wybranych % (kod 3). Oba udziały przeciągnij do tabeli „Według kategorii”.

4 We fragmentatorze zaznacz (Ctrl + klik) Komputery, Telefony oraz Telewizory i wideo. Który udział się zmienił i dlaczego?

1  Sprzedaż online PLN = CALCULATE([Sprzedaż PLN], 'Kanał'[Typ kanału] = "Online")

Udział online % = DIVIDE([Sprzedaż online PLN], [Sprzedaż PLN])

2  Sprzedaż wszystkich produktów PLN = CALCULATE([Sprzedaż PLN], REMOVEFILTERS('Produkt'))

Udział w całości % = DIVIDE([Sprzedaż PLN], [Sprzedaż wszystkich produktów PLN])

3  Udział w wybranych % = DIVIDE([Sprzedaż PLN], CALCULATE([Sprzedaż PLN], ALLSELECTED('Produkt')))

Rezultat ćwiczenia 2

SPRAWDŹ U SIEBIE

• Komputery: 29,2% całości, 
43,9% wybranych

• suma udziałów w całości: 
66,4%, w wybranych: 100%

• online: 41,8% → 50,8% → 
56,2%

• KEEPFILTERS: liczba tylko 
w wierszu Komputery

Plik INWD_dax_rozwiazanie.pbip, strona „Ćw. 2”. Wykres udziału online nie reaguje na fragmentator kategorii (Edytuj interakcje).

04
Iteratory: licz wiersz po wierszu



SUMX: najpierw każdy wiersz, potem suma
SUM dodaje gotową kolumnę. SUMX liczy wyrażenie w każdym wierszu tabeli i dopiero wyniki sumuje.

Ilość Cena jednostkowa Ilość × Cena Wartość sprzedaży

13 201,47 2 619,11 2 488,15

21 177,33 3 723,93 3 723,93

28 94,13 2 635,64 2 635,64

… … … …

Suma 874 897 885 850 573 496

SUMX('Sprzedaż', 'Sprzedaż'[Ilość] * 'Sprzedaż'[Cena jednostkowa])

SUM('Sprzedaż'[Wartość sprzedaży])

RÓŻNICA TO RABATY

• 24,3 mln zł, czyli 2,8% wartości 
brutto.

• W 2024 r. rabaty zjadły 4,8% —
w 2023 r. 1,8%.

• Pierwsze trzy wiersze to prawdziwe 
wiersze z pliku sprzedaz.csv.

Przejście kontekstu: miara w iteratorze działa, samo SUM nie
Średnia dzienna sprzedaż w 2024 r. Trzy wersje kodu, dwa wyniki.

AVERAGEX(VALUES('Kalendarz'[Data]),

[Sprzedaż PLN])

DOBRZE 715 937 zł

odwołanie do miary to ukryty CALCULATE: bieżąca data staje się filtrem

AVERAGEX(VALUES('Kalendarz'[Data]),

SUM('Sprzedaż'[Wartość sprzedaży]))

ŹLE 262 032 917 zł

SUM nie widzi wiersza: 366 razy liczy cały rok

AVERAGEX(VALUES('Kalendarz'[Data]),

CALCULATE(SUM('Sprzedaż'[Wartość sprzedaży])))

DOBRZE 715 937 zł

CALCULATE zamienia kontekst wiersza na filtr — to samo co miara

Klasyk podpowiedzi AI: kod wygląda rozsądnie i zwraca liczbę. Tylko liczba jest 366 razy za duża.

RANKX i ISINSCOPE: ranking bez bzdury w wierszu sumy

Ranking podkategorii =

IF(

ISINSCOPE('Produkt'[Podkategoria]),

RANKX(ALLSELECTED('Produkt'[Podkategoria]),

[Sprzedaż PLN])

)

Podkategoria Sprzedaż PLN bez ISINSCOPE z ISINSCOPE

Laptopy 185 933 469 1 1

Telewizory 150 764 702 2 2

Smartfony 134 239 218 3 3

… 22 kolejne

Suma 850 573 496 1

• RANKX porównuje sprzedaż podkategorii 
z każdą z listy ALLSELECTED — z tymi, które 
widać w tabeli.

• Bez warunku wiersz sumy dostaje miejsce 1: 
suma jest większa od każdej podkategorii.

• ISINSCOPE sprawdza, czy tabela grupuje po 
podkategorii. W sumie nie grupuje, więc miara 
zwraca pusto.

Ćw. 3 · Iteratory i ranking 15 min

1 Sprzedaż brutto PLN = SUMX('Sprzedaż', 'Sprzedaż'[Ilość] * 'Sprzedaż'[Cena jednostkowa])

2 Udział rabatów % = DIVIDE([Sprzedaż brutto PLN] - [Sprzedaż PLN], [Sprzedaż brutto PLN])

3 Średnia dzienna sprzedaż PLN = AVERAGEX(VALUES('Kalendarz'[Data]), [Sprzedaż PLN])

4 Ranking podkategorii z kodu poniżej wstaw do tabeli „Według podkategorii”.

Ranking podkategorii = IF(ISINSCOPE('Produkt'[Podkategoria]),

RANKX(ALLSELECTED('Produkt'[Podkategoria]), [Sprzedaż PLN]))

PYTANIE

Rabaty w 2024 r. urosły 2,6 raza. Które kategorie za to odpowiadają? Sprawdź na stanowisku.

Rezultat ćwiczenia 3

SPRAWDŹ U SIEBIE

• rabaty: 1,8% → 4,8% → 1,9%

• średnia dzienna: 780 tys. → 
716 tys. → 832 tys. zł

• ranking: Laptopy, Telewizory, 
Smartfony

• wiersz sumy rankingu: pusty

Plik INWD_dax_rozwiazanie.pbip, strona „Ćw. 3”.

05
Pięć funkcji, które robią z liczby komunikat



Pięć funkcji na jednej stronie raportu

Funkcja Co robi Miara w pliku Wynik dla 2024

SELECTEDVALUE zwraca wartość, gdy filtr zostawia dokładnie 
jedną

Tytuł dynamiczny 2024

SWITCH(TRUE()) sprawdza warunki po kolei, bierze pierwszy 
prawdziwy

Segment podkategorii Laptopy: Lider

TOPN zwraca N wierszy z najwyższą wartością Top 3 podkategorie 3 podkategorie

CONCATENATEX skleja wartości z tabeli w jeden tekst Top 3 podkategorie Laptopy, Smartfony, Telewizory

UNICHAR i FORMAT wstawia znak Unicode, liczbę zapisuje jako tekst Trend r/r ▼ -8,0%

Każda zwraca tekst albo jedną wartość. Wstawisz ją w tytuł, kartę albo formatowanie warunkowe.

SELECTEDVALUE: tytuł, który wie, co wybrałeś

Tytuł dynamiczny =

VAR _okres =

SELECTEDVALUE('Kalendarz'[Rok], "w wybranych latach")

VAR _zmiana = [Zmiana r/r %]

RETURN

"Sprzedaż " & _okres & ": "

& FORMAT([Sprzedaż PLN] / 1000000, "#,0.0")

& " mln zł"

& IF(NOT ISBLANK(_zmiana),

" (" & FORMAT(_zmiana, "+0.0%;-0.0%") & " r/r)")

Fragmentator Rok Tytuł wykresu

2024 Sprzedaż 2024: 262,0 mln zł (-8,0% r/r)

2025 Sprzedaż 2025: 303,7 mln zł (+15,9% r/r)

2024 i 2025 Sprzedaż w wybranych latach: 565,8 mln zł

nic Sprzedaż w wybranych latach: 850,6 mln zł

JAK TO PODPIĄĆ

• Zaznacz wykres → Formatowanie → Ogólne 
→ Tytuł → fx.

• Styl formatu: Wartość pola. Pole: Tytuł 
dynamiczny.

• Drugi argument SELECTEDVALUE to tekst na 
wypadek kilku lat albo braku wyboru.

• Ta sama funkcja liczy Sprzedaż rok wcześniej: 
rok z filtra minus jeden. Na zajęciach 5 zrobisz 
to funkcjami analizy czasowej.

SWITCH(TRUE()): warunki po kolei, wygrywa pierwszy prawdziwy

Segment podkategorii =

VAR _udzial = [Udział w całości %]

RETURN

SWITCH(

TRUE(),

NOT HASONEVALUE('Produkt'[Podkategoria])

|| ISBLANK(_udzial), BLANK(),

_udzial >= 0.10, "Lider",

_udzial >= 0.03, "Środek",

"Ogon"

)

• Warunki czytasz z góry na dół. Dlatego próg 
10% stoi przed progiem 3%.

• Ostatni argument bez warunku znaczy „w 
każdym innym przypadku”.

• HASONEVALUE zostawia pusty wiersz sumy: 
suma nie jest podkategorią.

WYNIK 2023–2025

3 liderów, 6 w środku, 16 w ogonie. Liderzy dają 
55,4% sprzedaży.

TOPN i CONCATENATEX: lista liderów jako jeden tekst

Top 3 podkategorie =

CONCATENATEX(

TOPN(3, VALUES('Produkt'[Podkategoria]),

[Sprzedaż PLN]),

'Produkt'[Podkategoria],

", ",

[Sprzedaż PLN], DESC

)

• TOPN(3, …) wybiera trzy podkategorie 
o najwyższej sprzedaży w bieżącym kontekście.

• CONCATENATEX skleja je przecinkiem ze spacją 
i sortuje według sprzedaży malejąco.

Okres Top 3 podkategorie

2023–2025 Laptopy, Telewizory, Smartfony

2024 Laptopy, Smartfony, Telewizory

ZAUWAŻ

W 2024 r. smartfony wyprzedziły telewizory 
o 26 tys. zł — o 0,06%. Tekst z miary pokaże to od 
razu, słupki ledwo.

UNICHAR, FORMAT i kolor z miary: strzałka trendu w tabeli

Trend r/r =

VAR _zmiana = [Zmiana r/r %]

RETURN

IF(

NOT ISBLANK(_zmiana),

IF(_zmiana >= 0, UNICHAR(9650), UNICHAR(9660))

& " " & FORMAT(_zmiana, "+0.0%;-0.0%")

)

Kolor trendu =

VAR _zmiana = [Zmiana r/r %]

RETURN

SWITCH(TRUE(), ISBLANK(_zmiana), BLANK(),

_zmiana < 0, "#C0392B", "#18480C")

Kategoria, 2024 Trend r/r

Komputery ▼ -28,2%

Telewizory i wideo ▼ -13,9%

Telefony ▲ +5,3%

AGD ▲ +7,8%

Audio ▲ +16,1%

Smart home i wearables ▲ +30,4%

Aparaty i kamery ▼ -18,5%

Gry i konsole ▲ +29,6%

Kolor podepniesz tak: tabela → Formatowanie → Elementy 
komórki → Kolor czcionki → fx → Wartość pola → Kolor 
trendu.

Ćw. 4 · Pięć funkcji na stronie „Ćwiczenia” 20 min

1 Sprzedaż rok wcześniej PLN i Zmiana r/r % (kod na ściądze DAX).

2 Trend r/r i Kolor trendu. Tabela: Kategoria, Sprzedaż PLN, Trend r/r, kolor 
czcionki z miary.

3 Segment podkategorii w tabeli z podkategoriami i Udziałem w całości %.

4 Top 3 podkategorie na karcie.

5 Wykres słupkowy sprzedaży według kategorii z tytułem z miary. 
Fragmentator Rok ustaw na 2024.

Wskazówka: Kod jest też w pliku rozwiązania: _Miary → folder Ćw. 4 · pięć funkcji.

Cel: strona „Ćw. 4” z pliku rozwiązania.

Rezultat ćwiczenia 4

SPRAWDŹ U SIEBIE

• tytuł: Sprzedaż 2024: 262,0 
mln zł (-8,0% r/r)

• top 3: Laptopy, Smartfony, 
Telewizory

• Komputery ▼ -28,2%, Smart 
home i wearables ▲ +30,4%

• segmenty 2024: 3 liderów, 6 
w środku

Plik INWD_dax_rozwiazanie.pbip, strona „Ćw. 4”, fragmentator Rok = 2024.

Dla ambitnych: OFFSET i WINDOW, czyli poprzedni wiersz i okno
Funkcje okna liczą po wierszach posortowanej listy. Tu lista to 12 kwartałów.

OFFSET: KWARTAŁ WCZEŚNIEJ

Sprzedaż poprzedni kwartał PLN =

CALCULATE([Sprzedaż PLN],

OFFSET(-1,

ALLSELECTED('Kalendarz'[Rok i kwartał],

'Kalendarz'[Klucz kwartału]),

ORDERBY('Kalendarz'[Klucz kwartału], ASC)))

WINDOW: BIEŻĄCY I DWA POPRZEDNIE

Średnia z 3 kwartałów PLN =

AVERAGEX(

WINDOW(-2, REL, 0, REL,

ALLSELECTED('Kalendarz'[Rok i kwartał],

'Kalendarz'[Klucz kwartału]),

ORDERBY('Kalendarz'[Klucz kwartału], ASC)),

[Sprzedaż PLN])

• OFFSET(-1, …) i WINDOW(-2, REL, 0, REL, …) liczą pozycje względem bieżącego 
kwartału.

• Relacja zawiera obie kolumny, bo wykres grupuje po Rok i kwartał i po kolumnie 
sortowania.

2025 Q4

102,4 mln zł; kwartał wcześniej 73,4 mln 
(+39,5% k/k); średnia z 3 kwartałów 81,6 mln.

Ćw. 5 · Dla ambitnych: poprzedni kwartał i średnia krocząca 10 min

1 Utwórz Sprzedaż poprzedni kwartał PLN (OFFSET) i Zmiana k/k %.

2 Utwórz Średnia z 3 kwartałów PLN (WINDOW).

3 Wykres liniowy: oś Rok i kwartał, wartości Sprzedaż PLN i Średnia z 3 
kwartałów PLN.

4 Tabela: Rok i kwartał, Sprzedaż PLN, poprzedni kwartał i Zmiana k/k %.

PYTANIE

Pierwszy kwartał 2023 nie ma poprzednika, a średnia z 3 kwartałów ma 
wartość. Dlaczego? Strona „Dla ambitnych” z pliku rozwiązania.

06
DAX w 2026: trzy narzędzia, które warto znać



Widok zapytań DAX: sprawdź miarę bez budowania wykresu
Czwarta ikona na lewym pasku Power BI Desktop. Piszesz zapytanie, dostajesz tabelę wyników.

TEST MIAR NA MAŁYM WYCINKU

EVALUATE

SUMMARIZECOLUMNS(

'Kalendarz'[Rok],

"Marża %", [Marża %],

"Udział rabatów %", [Udział rabatów %]

)

SPIS WSZYSTKICH MIAR MODELU

EVALUATE INFO.VIEW.MEASURES()

• Zwraca nazwę, tabelę, folder, opis i kod każdej miary. 
W pliku rozwiązania — 27 wierszy.

• Gotowy spis do manifestu intencji i do sprawdzenia, czy 
każda miara ma opis.

DLACZEGO TO WAŻNE

Zapytanie pokazuje same liczby, bez wykresu, który mógłby coś ukryć. Trzy wiersze porównasz z tabelą na stanowisku testowym.

Funkcje użytkownika (UDF): napisz raz, użyj w wielu miarach
Od czerwca 2026 r. ogólnodostępne w Power BI Desktop i w usłudze. Model musi mieć poziom zgodności 1702 lub wyższy.

DEFINE

FUNCTION ZmianaProcent =

(teraz : NUMERIC, wczesniej : NUMERIC) =>

DIVIDE(teraz - wczesniej, wczesniej)

EVALUATE

SUMMARIZECOLUMNS(

'Kalendarz'[Rok],

"Zmiana r/r",

ZmianaProcent([Sprzedaż PLN],

[Sprzedaż rok wcześniej PLN])

)

• Na naszym modelu zwraca -8,0% dla 2024 
i +15,9% dla 2025 — tak samo jak miara 
Zmiana r/r %.

• Zapis do modelu: przycisk nad definicją 
w widoku zapytań DAX (po angielsku: Update 
model: Add new function).

• W projekcie PBIP funkcja trafia do pliku 
definition/functions.tmdl.

• Nazwa: litery, cyfry, podkreślenia i kropki, bez 
spacji.

Jedna funkcja zamiast dwóch prawie takich samych miar: ZmianaProcent policzy zmianę 
r/r i k/k. Poprawiasz ją w jednym miejscu.

Źródło: Microsoft Learn, „Use DAX user-defined functions” (stan na 18.08.2026). Zapytanie przetestowane w Power BI Desktop 2.157 na pliku INWD_dax_rozwiazanie.

Obliczenia wizualne: DAX na wykresie, nie w modelu
Zaznacz wykres → Narzędzia główne → Nowe obliczenie wizualne. Formuła liczy na tym, co widać na wykresie.

Średnia krocząca = MOVINGAVERAGE([Sprzedaż PLN], 3)

Narastająco = RUNNINGSUM([Sprzedaż PLN])

Zmiana k/k = [Sprzedaż PLN] - PREVIOUS([Sprzedaż PLN])

ZYSKUJESZ

• krótszy kod: MOVINGAVERAGE zamiast WINDOW 
z ALLSELECTED i ORDERBY

• często szybsze liczenie: formuła pracuje na danych już 
zagregowanych

• gotowe szablony: suma narastająca, średnia krocząca, 
porównanie z poprzednim

TRACISZ

• obliczenie widzi tylko pola z tego wykresu

• nie użyjesz go w innej wizualizacji ani w mierze

• nie działa m.in. we fragmentatorach i w kluczowych 
czynnikach wpływu

Źródło: Microsoft Learn, „Visual calculations overview” (28.08.2026) i opis funkcji MOVINGAVERAGE.

07
AI pisze DAX. Ty sprawdzasz



Pięć kroków kontroli miary
Ta sama procedura dla miary z podręcznika, z forum i z czatu AI.

01

Przeczytaj

co filtruje, co iteruje, co 
zwraca — na głos

02

Przewidź

wynik jednej komórki, zanim 
uruchomisz

03

Zapytaj

EVALUATE na małym wycinku: 
jeden rok, jedna kategoria

04

Sprawdź brzegi

wiersz sumy, puste dane, 
dzielenie przez zero

05

Poruszaj filtrem

czy wynik reaguje tak, jak 
zapisałeś w manifeście

W dzienniku decyzji zapisz wynik kontrolny, np. „Marża % 2024 = 25,9% — zgodne z ręcznym przeliczeniem 67,9 mln / 262,0 mln”.

Typowe błędy w DAX-ie z czatu AI

Objaw Przyczyna Poprawka

kod woła 'Sales'[Amount] AI nie zna twojego modelu i zgaduje nazwy wklej nazwy tabel i kolumn z widoku modelu

średnia dzienna = sprzedaż całego roku SUM w iteratorze zamiast miary: brak przejścia 
kontekstu

odwołaj się do miary albo owiń wyrażenie w 
CALCULATE

udział nie reaguje na fragmentator REMOVEFILTERS tam, gdzie pytanie wymaga 
ALLSELECTED

wybierz mianownik świadomie

CALCULATE(…, FILTER(ALL('Produkt'), …)) stary wzorzec: wolniejszy, zdejmuje wszystkie filtry 
produktu

prosty filtr kolumny, w razie potrzeby KEEPFILTERS

błąd przy zerze w mianowniku dzielenie operatorem / DIVIDE(a, b)

COUNTIFS, IFS, VLOOKUP w mierze funkcje z Excela — DAX ich nie ma CALCULATE(COUNTROWS(…)), SWITCH(TRUE()), 
RELATED

Trzy środkowe błędy nie dają komunikatu o błędzie, tylko zwracają liczbę. Dlatego potrzebujesz kroku „przewidź”.

Na obronie: karty A1–A4 to dzisiejsze wzorce

Karta Zadanie na obronie Wzorzec z dziś

A1 · Udział w całości udział głównej miary w sumie dla wszystkich 
kategorii

REMOVEFILTERS i DIVIDE — ćw. 2

A2 · Rok do roku zmiana r/r w procentach na karcie albo w tabeli SELECTEDVALUE i CALCULATE — ćw. 4

A3 · Pierwsza piątka 5 najlepszych pozycji według głównej miary RANKX z ISINSCOPE albo TOPN — ćw. 3 i 4

A4 · Średnia na jednostkę średnia na klienta, produkt albo dzień z DIVIDE DIVIDE i AVERAGEX — ćw. 1 i 3

Na obronie zmieniasz swój projekt, nie plik ćwiczeniowy. Przećwicz każdy wzorzec na własnych danych przed CP2.

Na następne zajęcia

• CP2 na zajęciach 5: model i miary twojego projektu jako PBIP — co najmniej 3 nietrywialne miary

• Załóż w projekcie tabelę _Miary i przenieś do niej wzorce: udział, średnią, ranking, tekst z miary

• Każdą miarę sprawdź zapytaniem EVALUATE i zapisz wynik kontrolny w dzienniku decyzji

• Pisałeś DAX z AI? Wpisz to w deklarację: narzędzie, prompt, co przyjąłeś, co poprawiłeś

• Przeczytaj: M. Russo, A. Ferrari, Kompletny przewodnik po DAX, rozdz. 4–5 (konteksty i CALCULATE)

Zajęcia 5: analiza czasowa (SAMEPERIODLASTYEAR, DATESYTD), wydajność miar i publikacja raportu.

UEP · Informatyczne narzędzia wizualizacji danych  ·  dr inż. Szczepan Górtowski 45



Trzy rzeczy do zapamiętania

kontekst
miara liczy się w filtrach komórki — także w 

wierszu sumy

CALCULATE
dodaje, usuwa albo zostawia filtry: to jego trzy 

ruchy

przewidź
zanim uwierzysz miarze, zgadnij jej wynik dla 

jednej komórki

